

# **Sun's position for navigation with DM15/HP-15c Manual**

Michael Josefsson

October 14, 2023



# Contents

1.1	Overview . . . . .	4
1.2	General notes . . . . .	4
1.3	Public routines . . . . .	5
1.4	Usage for Sun . . . . .	5
1.4.1	First set AP ( <i>Assumed Position</i> ) . . . . .	5
1.4.2	Then set the date . . . . .	6
1.4.3	Finally enter time in GMT for Altitude and Azimuth . . . . .	6
1.4.4	Calculate Sun's GHA and Declination . . . . .	7
1.5	SHA and Declination → Altitude and Azimuth . . . . .	7
1.6	Date and time → GHA $\Upsilon$ and LHA $\Upsilon$ . . . . .	8
1.7	GHA and Declination → Altitude and Azimuth . . . . .	8
1.8	Use as <i>Sight Reduction Table</i> . . . . .	9
1.9	Program and information . . . . .	9
1.9.1	Register usage . . . . .	9
1.9.2	Program installation . . . . .	9
1.9.3	Program listing . . . . .	10

## 1.1 Overview

The handheld calculator DM15 (a HP-15c look-a-like with more memory) can be used for determining the sun's position with precision enough for celestial navigation purposes. The accompanying program, listed in the Appendix, constitutes a handy tool for either finding the *Nautical Almanac's* entries GHA (*Greenwich Hour Angle*) and Declination, or — with AP (*Assumed Position*) — directly calculate the sun's Altitude  $H_c$  and Azimuth  $A_z$  for this position.

The algorithm relies on pure Keplerian motion of the sun. No planetary perturbations are taken into account. Resulting angular accuracy is about 1 minute of arc, which is adequate for general navigation at sea.

## 1.2 General notes

Before use, notice that:

- All times entered are UT ("GMT") even if observer's longitude is not the prime meridian. Of course, local hour angles take longitude into consideration, but all times are still UT. Time format is  $hh.mmss$ , where  $mm$  and  $ss$  must be two-digit numbers.
- The program makes use of the calculator's internal decimal to degrees, minutes and seconds routines both for **entry** and **displayed result**. In navigation a more common format of degrees, minutes and tenths of a minute is used. That conversion, if needed, is readily done by dividing the arc-seconds number or multiplying the minute's decimal by 6.

### Example

Convert angle in  $ddd.mmss$  to  $ddd.mm-t$

$98^\circ 26' 12''$ , entered as **98.2612**, is  $98^\circ 26.2'$  where  $12''/6 = 2$

$98^\circ 26' 43''$ , entered as **98.2643**, is  $98^\circ 26.7'$  where  $43''/6 \approx 7$

■

### Example

Convert angle in  $ddd.mm-t$  to  $ddd.mmss$

$14^\circ 7.3'$  is  $14^\circ 7' 18''$  where  $3 \cdot 6 = 18$

$277^\circ 4.5'$  is  $277^\circ 4.30'$  where  $5 \cdot 6 = 30$

■

## 1.3 Public routines

Full use of the program entails the following routines. Their use is described in this document.

Button	Function
<b>A</b>	Date for $\Upsilon$ hour angle at UT=0h
<b>B</b>	Time for Sun Altitude and Azimuth
<b>C</b>	SHA and declination for <i>own object</i>
<b>D</b>	GMT $\rightarrow$ <i>own object's</i> Altitude and Azimuth
<b>E</b>	GMT $\rightarrow$ GHA $\Upsilon$ and LHA $\Upsilon$
<b>.0</b>	GHA, time and declination $\rightarrow$ Altitude and Azimuth
<b>.5</b>	After <b>B</b> $\rightarrow$ GHA and declination (mimicks a <i>Nautical Almanac</i> entry)

## 1.4 Usage for Sun

The program's main purpose is to give altitude and azimuth for the sun. This requires an *Assumed Position*, date and time.

### 1.4.1 First set AP (*Assumed Position*)

An AP is entered in registers **8** and **.8** before any calculations can be performed.

#### Example

Entering AP.

A location of Lat N58° 34', Long E14° 34' 12" is entered into registers **8** and **.8**:

Data	Format	Key	Display shows	
<b>58.3400</b>	$\pm dd.mm.ss$	g ->H	<b>STO 8</b>	<b>58.5667</b>
<b>14.3412</b>	$\pm dd.mm.ss$	g ->H	<b>STO .8</b>	<b>14.5700</b>

East and North are positive, West and South are negative.

The position is permanently stored until manually changed and need only be set once.

■

## 1.4.2 Then set the date

Each day has its own parameters and require the **A**-routine to be run each new day.

### Example

Entering the date.

Enter June 12th 2022, i.e. year 2022, month 6 and day 12

Data	Format	Key	Display shows
2022	YYYY	ENTER	2022.0000
6	mm	ENTER	6.0000
12	dd	f <b>A</b>	260.1816 (260° 18' 17" = 260° 18.3', GHA $\Upsilon$ at 0h)

This is only necessary once for each day.

■

## 1.4.3 Finally enter time in GMT for Altitude and Azimuth

Next the sun's coordinates for time of date UT/GMT can be calculated.

### Example

Find sun's Hc and Az for UT 09h 54m 48s. Date entered as above.

Enter time in format *hh.mmss* then use routine **B**.

Data	Format	Key	Display shows
9.5448	hh.mmss	f <b>B</b>	52.3845 ( Hc = 52° 38' 45" )
		x<>y	154.1140 ( Az = 154° 11' 40" )

**Result:** Hc = 52° 38.7', Az = 154°

A new time can be entered directly. For example, also find sun's Hc and Az a few minutes later at UT 10h 02m 30s.

Data	Format	Key	Display shows
10.0230	hh.mmss	f <b>B</b>	53.0338 ( Hc = 53° 03' 38" )
		x<>y	157.0236 ( Az = 157° 2' 36" )

**Result:** Hc = 53° 03.6', Az = 157°

■

### 1.4.4 Calculate Sun's GHA and Declination

If wanted the program can also produce values for GHA and Declination imitating the *Nautical Almanac*.

#### Example

Find GHA and decl for 10h on June 12th 2023

The date must be set as above. Now enter time to get Hc and Az if wanted. Or ignore the output and use **GSB .5** to get GHA and decl:

Data	Format	Key	Display shows
10.0000	hh.mmss	f B	52.5508 ( Hc = 52° 55' 08", ignore )
		GSB .5	330.0248 ( GHA = 330° 2' 48" )
		x<>y	23.0901 ( Decl = N23° 09' 01" )

**Result:** GHA = 330° 2.8', Decl = N23° 09.0' (*Nautical Almanac* gives 330° 2.8' and N23° 8.8').

GHA and Decl can of course be calculated for any other time during the day in the same manner.

## 1.5 SHA and Declination → Altitude and Azimuth

The coordinates of a celestial object, for example a star, are given as SHA (*Sidereal Hour Angle*) and Declination.<sup>1</sup>

#### Example

Coordinates of *Vega* (SHA 80° 34.3', declination 38° 48.2') are entered

Data	Format	Key	Display shows
80.3418	ddd.mmss	ENTER	80.3418
38.4812	ddd.mmss	f C	279.4283 (RA in decimal degrees as a bonus)

Now find *Vega*'s calculated position for UT = 23h 02m 10s on June 12th 2023 already entered above.

Data	Format	Key	Display shows
23.0210	hh.mmss	f D	67.0015 ( Hc = 67° 00' 15" )
		x<>y	141.1224 ( Az = 141° 12' 24" )

**Result:** Vega can be expected at Hc = 67° 0.2' and Zn = 141°. Set the sextant for 67° and search for it in south-east.

<sup>1</sup>Right Ascension can be entered as  $\alpha = 360 - SHA$  if needed.

## 1.6 Date and time → GHA $\Upsilon$ and LHA $\Upsilon$

Find GHA  $\Upsilon$  on 4 October 2022 at 7h 57m 20s. Also find LHA  $\Upsilon$  longitude in .8 (E14° 34' 12" as before).

Data	Format	Key	Display shows
<b>2022</b>	YYYY	ENTER	<b>2022.0000</b>
<b>10</b>	mm	ENTER	<b>10.0000</b>
<b>4</b>	dd	f <b>A</b>	<b>12.4006</b> (12° 40' 6", GHA $\Upsilon$ at 0h)
<b>7.5720</b>	hh.mmss	f <b>E</b>	<b>132.1942</b> ( GHA $\Upsilon$ = 132° 19' 42" )
		<b>x&lt;&gt;y</b>	<b>146.5354</b> ( LHA $\Upsilon$ = 146° 53' 54" )

## 1.7 GHA and Declination → Altitude and Azimuth

When coordinates are given as GHA (*Greenwich Hour Angle*) and Declination, as is the case for the planets and Moon listed in NA, proceed as follows. As GHA changes continuously during the day a time is also required.

### Example

Coordinates of *Sun* as from *Nautical Almanac* above are GHA 330° 2.8', declination N23° 8.8' at 10h.

Data	Format	Key	Display shows
<b>330.0248</b>	ddd.mmss	<b>ENTER</b>	<b>330.0248</b>
<b>10.0000</b>	ddd.mmss	<b>GSB .0</b>	<b>279.3413</b> , ignore
<b>23.0848</b>	ddd.mmss	<b>R/S</b>	<b>52.5455</b> ( Hc = 52° 54' 55" )
		<b>x&lt;&gt;y</b>	<b>156.0821</b> ( Az = 156° 08' 21" )

**Result:** The *Sun* is expected at a height of 52° 54.9' and in a direction of 156°. <sup>2</sup>

■

<sup>2</sup>Compare these results with the direct method above (52° 54' 55" vs 52° 55' 08" and 156° 08' 16" vs 156° 08' 21"). Pretty close but not exactly the same.



## 1.8 Use as *Sight Reduction Table*

The program can also solve the navigational triangle and be used as a (Ho-214/Ho-229 etc) *Sight Reduction Table* replacement. To solve the triangle AP latitude, object's declination and hour angle needs to be entered.

AP latitude is entered in register **8** as before, declination is set via **C** and hour angle is entered into register **.2**. The hour angle is positive if westward.

### Example

Find Hc and Az as in Ho-214

Assume longitude  $58^{\circ}$  N, Declination  $8^{\circ} 30'$  and an hour angle of  $54^{\circ}$  (object to the west of observer).

Data	Format	Key	Display shows
<b>58</b>	<i>dd.mmss</i>	<b>STO 8</b>	<b>58.0000</b>
<b>8.3000</b>	<i>dd.mmss</i>	<b>C</b>	<b>8.5000</b>
<b>54</b>	<i>dd.mmss</i>	<b>g -&gt;H</b>	<b>54.0000</b>
		<b>STO .2</b>	<b>54.0000</b>
		<b>GSB 7</b>	<b>25.4102 ( Hc = <math>25^{\circ} 41' 02''</math>)</b>
		<b>x&lt;&gt;y</b>	<b>242.3616 ( Az = <math>242^{\circ} 36' 16'' = 242.60^{\circ}</math>)</b>

Ho-214 gives Alt. =  $25^{\circ} 41.0'$  and Az. =  $117.4^{\circ}$ . Where true azimuth is  $360 - 117.4 = 242.6^{\circ}$ .

## 1.9 Program and information

### 1.9.1 Register usage

The lower registers **r0..r7** are used by the calculator's statistics functions and are not *permanently* used by this program. They *are* used however for intermediate results via the normal operating sequences **A-B** or **A-C-D** or **A-E**.

In short: Use **r0..r7** as you wish but they will be altered by **A**.

### 1.9.2 Program installation

For a fresh install of the program perform steps 1–6 below.

## Contents

1. Make space on the DM15 for program and registers:
  - Enter **21 f DIM (i)**
  - Double check: **g MEM** should read **21.209**
2. In HP-15C/Preferences/DM15 menu: Select 229 as Number of registers.
3. File/Open Program: file.15c
4. Write program to DM15.
  - On device enable serial communication (hold C while pressing ON-button)
  - File/Write DM15
5. Before use a number of constants must be entered in the following registers:

Register	Constant
.3	279.4055638
.4	283.3328093
.5	1.016860112
.6	23.44188400
.7	0.002737909, (1/365.2422)
6. That's it. Now the samples in this document should give the expected results.

### 1.9.3 Program listing

Note: In the listing below some minor self explanatory key appearances have changed.  $\text{SIN}^{-1}$  is replaced with **ASIN** etc,  $\mathbf{x} \leftrightarrow \mathbf{y}$  is  **$\mathbf{x} <> \mathbf{y}$**  and  **$\mathbf{R} \downarrow$**  is **Rv**.

```
000 {      }
001 { 42 21 48 8 } f LBL .8
002 {      3 } 3
003 {      6 } 6
004 {      0 } 0
005 {      43 32 } g RTN
006 { 42 21 4 } f LBL 4
007 {      23 } SIN
008 {      34 }  $\mathbf{x} <> \mathbf{y}$ 
009 {      23 } SIN
010 { 22 48 6 } GTO .6
011 { 42 21 5 } f LBL 5
012 {      24 } COS
013 {      34 }  $\mathbf{x} <> \mathbf{y}$ 
014 {      24 } COS
015 { 22 48 6 } GTO .6
016 { 42 21 2 } f LBL 2
017 { 32 48 8 } GSB .8
018 {      10 } /
019 { 42 44 } f FRAC
020 { 43 30 1 } g TEST  $\mathbf{x} > 0$ 
021 { 22 3 } GTO 3
022 { 1 } 1
023 { 40 } +
```

## 1.9 Program and information

```

024 {      42 21 3 } f LBL 3
025 {      32 48 8 } GSB .8
026 { 42 21 48 6 } f LBL .6
027 {      20 } *
028 {      43 32 } g RTN
029 { 42 21 48 2 } f LBL .2
030 {      1 } 1
031 {      5 } 5
032 {      22 48 6 } GTO .6
033 {      42 21 12 } f LBL B
034 {      32 15 } GSB E
035 {      45 6 } RCL 6
036 {      2 } 2
037 {      4 } 4
038 {      5 } 5
039 {      9 } 9
040 {      9 } 9
041 {      4 } 4
042 {      4 } 4
043 {      48 } .
044 {      5 } 5
045 {      30 } -
046 {      45 4 } RCL 4
047 {      2 } 2
048 {      4 } 4
049 {      10 } /
050 {      40 } +
051 {      45 48 7 } RCL .7
052 {      32 48 8 } GSB .8
053 {      20 } *
054 {      20 } *
055 {      45 48 3 } RCL .3
056 {      40 } +
057 {      45 48 4 } RCL .4
058 {      30 } -
059 {      42 3 } f ->RAD
060 {      44 9 } STO 9
061 {      43 8 } g RAD
062 {      36 } ENTER
063 {      1 } 1
064 {      0 } 0
065 {      42 7 9 } f FIX 9
066 {      42 10 8 } f SOLVE 8
067 {      42 7 4 } f FIX 4
068 {      2 } 2
069 {      10 } /
070 {      25 } TAN
071 {      45 48 5 } RCL .5
072 {      20 } *
073 {      43 25 } g ATAN
074 {      2 } 2
075 {      20 } *
076 {      43 3 } g ->DEG
077 {      43 7 } g DEG
078 {      45 48 4 } RCL .4
079 {      40 } +
080 {      44 48 0 } STO .0
081 {      45 48 0 } RCL .0

082 {      23 } SIN
083 {      45 48 6 } RCL .6
084 {      24 } COS
085 {      20 } *
086 {      45 48 0 } RCL .0
087 {      24 } COS
088 {      43 1 } g ->P
089 {      33 } Rv
090 {      44 9 } STO 9
091 {      45 48 0 } RCL .0
092 {      23 } SIN
093 {      45 48 6 } RCL .6
094 {      23 } SIN
095 {      20 } *
096 {      43 23 } g ASIN
097 {      44 48 0 } STO .0
098 {      45 9 } RCL 9
099 { 42 21 48 1 } f LBL .1
100 {      45 5 } RCL 5
101 {      32 48 8 } GSB .8
102 {      45 9 } RCL 9
103 {      30 } -
104 {      40 } +
105 {      32 2 } GSB 2
106 {      44 48 2 } STO .2
107 {      32 7 } GSB 7
108 {      43 32 } g RTN
109 {      42 21 7 } f LBL 7
110 {      45 48 2 } RCL .2
111 {      24 } COS
112 {      45 8 } RCL 8
113 {      45 48 0 } RCL .0
114 {      32 5 } GSB 5
115 {      20 } *
116 {      45 48 0 } RCL .0
117 {      45 8 } RCL 8
118 {      32 4 } GSB 4
119 {      40 } +
120 {      43 23 } g ASIN
121 {      36 } ENTER
122 {      36 } ENTER
123 {      44 48 1 } STO .1
124 {      45 8 } RCL 8
125 {      32 4 } GSB 4
126 {      16 } CHS
127 {      45 48 0 } RCL .0
128 {      23 } SIN
129 {      40 } +
130 {      34 } x<>y
131 {      45 8 } RCL 8
132 {      32 5 } GSB 5
133 {      10 } /
134 {      43 24 } g ACOS
135 { 42 4 48 2 } f X<>.2
136 {      23 } SIN
137 {      43 30 2 } g TEST x<0
138 {      22 6 } GTO 6
139 {      32 48 8 } GSB .8

```

## Contents

140 {	45 48 2 }	RCL .2	198 {	36 }	ENTER
141 {	30 }	-	199 {	0 }	0
142 {	44 48 2 }	STO .2	200 {	32 1 }	GSB 1
143 {	42 21 6 }	f LBL 6	201 {	2 }	2
144 {	45 48 1 }	RCL .1	202 {	4 }	4
145 {	45 48 2 }	RCL .2	203 {	1 }	1
146 {	22 9 }	GTO 9	204 {	5 }	5
147 {	42 21 8 }	f LBL 8	205 {	0 }	0
148 {	23 }	SIN	206 {	2 }	2
149 {	48 }	.	207 {	0 }	0
150 {	0 }	0	208 {	30 }	-
151 {	1 }	1	209 {	3 }	3
152 {	6 }	6	210 {	6 }	6
153 {	7 }	7	211 {	5 }	5
154 {	1 }	1	212 {	2 }	2
155 {	8 }	8	213 {	5 }	5
156 {	20 }	*	214 {	10 }	/
157 {	16 }	CHS	215 {	36 }	ENTER
158 {	40 }	+	216 {	36 }	ENTER
159 {	45 9 }	RCL 9	217 {	48 }	.
160 {	30 }	-	218 {	0 }	0
161 {	43 32 }	g RTN	219 {	0 }	0
162 {	42 21 13 }	f LBL C	220 {	0 }	0
163 {	43 2 }	g ->H	221 {	0 }	0
164 {	44 48 0 }	STO .0	222 {	2 }	2
165 {	33 }	Rv	223 {	5 }	5
166 {	43 2 }	g ->H	224 {	8 }	8
167 {	32 48 8 }	GSB .8	225 {	1 }	1
168 {	34 }	x<>y	226 {	20 }	*
169 {	30 }	-	227 {	2 }	2
170 {	44 9 }	STO 9	228 {	4 }	4
171 {	43 32 }	g RTN	229 {	0 }	0
172 {	42 21 15 }	f LBL E	230 {	0 }	0
173 {	43 2 }	g ->H	231 {	48 }	.
174 {	44 4 }	STO 4	232 {	0 }	0
175 {	45 48 7 }	RCL .7	233 {	5 }	5
176 {	1 }	1	234 {	1 }	1
177 {	40 }	+	235 {	2 }	2
178 {	20 }	*	236 {	6 }	6
179 {	32 48 2 }	GSB .2	237 {	2 }	2
180 {	45 1 }	RCL 1	238 {	40 }	+
181 {	40 }	+	239 {	20 }	*
182 {	36 }	ENTER	240 {	6 }	6
183 {	36 }	ENTER	241 {	48 }	.
184 {	45 48 8 }	RCL .8	242 {	6 }	6
185 {	40 }	+	243 {	4 }	4
186 {	32 2 }	GSB 2	244 {	6 }	6
187 {	44 5 }	STO 5	245 {	0 }	0
188 {	34 }	x<>y	246 {	6 }	6
189 {	32 2 }	GSB 2	247 {	5 }	5
190 {	34 }	x<>y	248 {	6 }	6
191 {	42 21 9 }	f LBL 9	249 {	40 }	+
192 {	42 2 }	f ->H.MS	250 {	32 48 2 }	GSB .2
193 {	34 }	x<>y	251 {	32 2 }	GSB 2
194 {	42 2 }	f ->H.MS	252 {	44 3 }	STO 3
195 {	43 32 }	g RTN	253 {	45 6 }	RCL 6
196 {	42 21 0 }	f LBL 0	254 {	44 0 }	STO 0
197 {	1 }	1	255 {	43 32 }	g RTN

## 1.9 Program and information

```

256 {      42 21 11 } f LBL A
257 {      44 4 } STO 4
258 {      33 } Rv
259 {      44 5 } STO 5
260 {      33 } Rv
261 {      32 0 } GSB 0
262 {      45 1 } RCL 1
263 {      45 5 } RCL 5
264 {      45 4 } RCL 4
265 {      32 1 } GSB 1
266 {      45 0 } RCL 0
267 {      30 } -
268 {      45 48 7 } RCL .7
269 {      32 48 8 } GSB .8
270 {      20 } *
271 {      20 } *
272 {      45 3 } RCL 3
273 {      40 } +
274 {      32 2 } GSB 2
275 {      44 1 } STO 1
276 {      42 2 } f ->H.MS
277 {      43 32 } g RTN
278 {      42 21 1 } f LBL 1
279 {      1 } 1
280 {      7 } 7
281 {      2 } 2
282 {      1 } 1
283 {      0 } 0
284 {      1 } 1
285 {      3 } 3
286 {      48 } .
287 {      5 } 5
288 {      40 } +
289 {      34 } x<>y
290 {      44 1 } STO 1
291 {      2 } 2
292 {      7 } 7
293 {      5 } 5
294 {      20 } *
295 {      9 } 9
296 {      10 } /
297 {      43 44 } g INT
298 {      40 } +
299 {      34 } x<>y
300 {      36 } ENTER
301 {      42 4 1 } f X<>1
302 {      9 } 9
303 {      40 } +
304 {      1 } 1

305 {      2 } 2
306 {      10 } /
307 {      43 44 } g INT
308 {      40 } +
309 {      7 } 7
310 {      20 } *
311 {      4 } 4
312 {      10 } /
313 {      43 44 } g INT
314 {      16 } CHS
315 {      40 } +
316 {      45 1 } RCL 1
317 {      3 } 3
318 {      6 } 6
319 {      7 } 7
320 {      20 } *
321 {      40 } +
322 {      44 6 } STO 6
323 {      43 32 } g RTN
324 {      42 21 14 } f LBL D
325 {      32 15 } GSB E
326 {      32 48 1 } GSB .1
327 {      43 32 } g RTN
328 {      42 21 48 5 } f LBL .5
329 {      45 5 } RCL 5
330 {      45 48 8 } RCL .8
331 {      30 } -
332 {      32 48 8 } GSB .8
333 {      45 9 } RCL 9
334 {      30 } -
335 {      40 } +
336 {      32 2 } GSB 2
337 {      45 48 0 } RCL .0
338 {      22 9 } GTO 9
339 {      42 21 48 0 } f LBL .0
340 {      32 15 } GSB E
341 {      43 2 } g ->H
342 {      43 33 } g R^
343 {      43 2 } g ->H
344 {      34 } x<>y
345 {      30 } -
346 {      42 2 } f ->H.MS
347 {      31 } R/S
348 {      32 13 } GSB C
349 {      45 4 } RCL 4
350 {      42 2 } f ->H.MS
351 {      32 14 } GSB D
352 {      43 32 } g RTN

```